CEMENT and CONCRETE RESEARCH. Vol. 4, pp. 797-806, 1974. Pergamon Press, Inc.
Printed in the United States.

COMPUTATION OF KELVIN CHAIN RETARDATION SPECTRA OF AGING CONCRETE

Zdeněk P. Bažant and Ali Asghari
Department of Civil Engineering
Northwestern University, Evanston, Illinois 60201 USA

(Communicated by F. H. Wittmann)
(Received May 2, 1974; in final form June 17, 1974)

ABSTRACT

A method of computing the age-dependent spectra of a Kelvin chain model from given data on creep of concrete is presented, along with a full listing of a FORTRAN IV subroutine. The method is based on a Dirichlet series expansion of creep rate as a function of the duration of creep test at constant current time. The inverse retardation moduli are approximated as a Dirichlet series in concrete age and as a quintic polynomial in the logarithm of the retardation time. These approximations serve at the same time as a smoothing device. They also enable explicit expressions for creep functions. The retardation moduli are determined by the method of least-squares. Numerical examples demonstrate that the retardation spectra completely characterize concrete creep because the original creep function can be recovered from them with a negligible error. The retardation spectra provide a rate-type creep law, which is necessary for analyzing large structural systems.

Une méthode de calcul des spectres de retardation du béton à des ages différents est présentée, avec un programme en FORTRAN IV. Le fluage linéaire est décrit par une chaîne de Kelvin dont les modules de ressorts $E_1,\dots,E_n$ dépendent de l'age. Les coefficients $E_1^{-1},\dots,E_n^{-1}$ sont introduits sous la forme d'une série d'exponentielles de l'age et d'un polynome quintic de la racine quintic du temps de retardation, et on les calcule par la méthode des plus petits carrés. Les exemples numériques montrent que les courbes de fluage données peuvent être obtenues à partir des spectres. Les spectre fournissent une loi de fluage en forme d'équations différentielles, ce qui est utile pour le calcul des constructions à grand nombre d'inconnues.

Introduction

Relaxation and retardation spectra are fundamental characteristics of decaying phenomena, having a similar significance as the frequency response spectra have for periodic phenomena. In the case of concrete creep, computation of the spectra is greatly complicated by a change of material properties with age, called aging. An efficient algorithm for computing the age-dependent relaxation spectra, corresponding to a Maxwell chain model, has been presented in a preceding paper (1). The aim of this paper is to develop a computational algorithm for the age-dependent retardation spectra, which correspond to the Kelvin (or Voigt) chain model (Fig. 1).



Fig. 1. Kelvin Chain Model and Retardation Spectrum.

A method for computing the Kelvin chain parameters from concrete creep data has already been presented in (2). However, in this method the elastic moduli of the Kelvin chain are obtained negative for certain periods of time. Although this fact does not preclude the use of Kelvin chain models in numerical creep analysis, it is physically unacceptable. A different method which is free of this drawback will be developed herein.

For a detailed discussion of the reasons for using the Kelvin or Maxwell chain models, the preceding papers may be consulted (1,2).

Mathematical Formulation

Without detracting from generality, attention may be restricted to uniaxial stress-strain relations. The Kelvin chain model (Fig. 1) then provides the following relations:

$$\varepsilon = \sum_{\mu=1}^{n} \varepsilon_\mu, \quad \dot{\varepsilon}_\mu = \frac{\dot{\sigma} - \dot{\sigma}_\mu}{E_\mu(t)}, \quad \sigma_\mu = \eta_\mu(t)\dot{\varepsilon}_\mu \quad (\mu = 1,\ldots,n), \tag{1}$$

where ε = strain, σ = stress; E_μ and η_μ are age-dependent spring moduli and dashpot viscosities of the μ-th Kelvin unit in the chain (Fig. 1); ε_μ = hidden strains = strain in the μ-th unit; σ_μ = stress in the μ-th dashpot; t = time = age of concrete measured from its casting; dots denote time derivatives, e.g. $\dot{\sigma} = d\sigma/dt$. It should be noted the elastic relation $\varepsilon_\mu = (\sigma - \sigma_\mu)/E_\mu$ is not equivalent to that in Eq.(1) and is thermodynamically impossible in the case of a hardening material (3). Because of this fact, the differential equations for the age-dependent Kelvin chain are more difficult to integrate than those for Maxwell chain (1).

The ratios $\eta_\mu(t)/E_\mu(t)$ are called retardation times τ_μ and the plot of $1/E_\mu$ versus log τ_μ is called retardation spectrum. In the preceding paper (1), it has been shown in detail that, for the Maxwell chain, the τ_μ-values may and must be chosen, provided they are sufficiently densely distributed in log-time and cover the whole range of interest (Fig. 1). The same reasons apply here.

In particular, τ_μ may be considered time-constant and, based on practical experience, a suitable choice is

$$\tau_1 \doteq 0, \quad \tau_\mu = 10^{\mu-2}\,\tau_2 \quad \text{for } \mu = 2,3,\ldots,n. \tag{2}$$

The zero retardation time is necessary to express instantaneous elastic deformations.

Excluding σ_μ, Eq.(1) may be rewritten as

$$\varepsilon = \sum_{\mu=1}^{n} \varepsilon_\mu, \quad \eta_\mu \ddot{\varepsilon}_\mu + (E_\mu + \dot{\eta}_\mu)\dot{\varepsilon}_\mu = \dot{\sigma}. \tag{3}$$

By virtue of the constancy of τ_μ, the latter differential equation may be easily integrated (2). Considering constant stress $\sigma = 1$ acting since time t' ($\leq$ t), the integration yields (2)

$$\dot{\varepsilon}(t) = \sum_\mu \frac{1}{\tau_\mu E_\mu(t)}\, e^{-(t-t')/\tau_\mu} \tag{4}$$

as may be verified by back substitution in Eq.(3). Eq.(4) is a series of real exponentials, called Dirichlet series. In contrast with the case of Maxwell chain (1), the coefficients of the series do not depend on t' but on t, which means that $1/E_\mu(t)$ can be obtained by expanding the curves of $\dot{\varepsilon}$ versus (t - t') at constant t (rather than constant t'). In a previous study of experimental data (2) it was found that these curves can have non-decaying portions (positive slope), which would mean that E_μ could be negative for a certain time period, and because negative E_μ is thermodynamically impossible the use of Eq. (4) was rejected. However, later it was proved (Eq. 3.33 in (3)) that the slope of the afore-mentioned curves must always be non-positive, and so the positive slope portions of the curves, when found from test data, are due to experimental error. There is thus no objection against Eq.(4). Moreover, if $\dot{\varepsilon}$ is a sufficiently smooth function of t and t', given creep data can be approximated by the Dirichlet series with any desired accuracy.*

If Eq.(4) were used for determining E_μ from test data, a plot of $\dot{\varepsilon}$ versus $(t-t')$ would have to be constructed from test data. This would involve differentiation of a random function, the experimental creep curve, and because the order of numerical error is always increased by numerical differentiation, the scatter of $\dot{\varepsilon}$ would be much higher than that of creep data. Therefore, it is imperative to integrate $\varepsilon(t)$ from Eq.(4). It has been found (2) that an explicit expression is possible if one assumes the dependence of $1/E_\mu$ upon t also in the form of Dirichlet series, i.e.,

$$\frac{1}{E_\mu(t)} = \sum_{k=1}^{m} J_{\mu k}\, e^{-t/\tau'_k} \quad (\mu = 1,2,\ldots,n) \tag{5}$$

where $J_{\mu k}$ are constants. This expression at the same time ensures that $E_\mu(t)$ is a smooth function of t. The τ'_k-values are suitably chosen as

$$\tau'_k = 10^{k-1}\,\tau'_1 \quad \text{for } k = 1,2,\ldots,m-1, \quad \tau'_m = \infty. \tag{6}$$

Integrating Eq.(4) and using Eq.(5) yields (4)

$$J(t,t') = \sum_{\mu=1}^{n}\sum_{k=1}^{m} J_{\mu k}\phi_{\mu k}(t,t'), \tag{7}$$

$$\phi_{\mu k}(t,t') = \frac{t-t'}{\tau_\mu}\,\frac{1-e^{-(t-t')/\tau_{\mu k}}}{(t-t')/\tau_{\mu k}}\,e^{-t'/\tau'_k}, \quad \tau_{\mu k} = \frac{\tau_\mu \tau'_k}{\tau_\mu + \tau'_k} \tag{8}$$

where $J(t,t')$ = creep function (or creep compliance) = strain at time t caused by a constant unit stress acting since time t'.

Identification of Material Parameters from Test Data

The problem of identifying the retardation spectrum from given test data is now reduced to fitting Eq.(7) to test data. Because this problem is equivalent to fitting Eq.(5) to the creep rate data, one may expect the same difficulties as those in fitting any test data with the Dirichlet series. It appears that $J_{\mu k}$ are unstable functions of test data, namely the same data can be fitted nearly equally well by very different sets of $J_{\mu k}$-values, i.e., the

*Note that $\partial\dot{\varepsilon}/\partial(t-t') = \partial^2 J(t,t')/\partial t\,\partial(t-t') = -\partial^2 J(t,t')/\partial t\partial t'$, in which one may substitute $-\partial J(t,t')/\partial t' = L(t,t') = \varepsilon$ at time t caused by a unit stress impulse (Dirac function) that occurred in time t'. The response to a stress impulse must always decay with time; hence $\partial L(t,t')/\partial t \le 0$ and $\partial\dot{\varepsilon}/\partial(t-t') \le 0$.

solution of the identification problem is not unique. To avoid the computational difficulties inherent to non-uniqueness, it is necessary to require that $1/E_\mu$ as a function of $\log \tau_\mu$ is sufficiently smooth, which is certainly also a physically natural assumption to make. The most convenient way of imposing a smoothing condition is to assume $1/E_\mu$ as a low order polynomial in $\log \tau_\mu$. Thus, in view of Eq.(4), it will be assumed that

$$J_{\mu k} = C_{1k} + C_{2k}\mu + C_{3k}\mu^2 + \cdots + C_{N+1\ k}\ \mu^N \text{ for } \mu > 1. \tag{9}$$

The value of $J_{\mu k}$ for $\mu = 1$ must be excluded from this smoothing polynomial because it does not characterize a single $1/E_\mu$-value but rather a sum of all $1/E_\mu$-values for which τ_μ is less than τ_2 (see Fig. 1). Equation (4) may now be compactly rewritten as

$$J_{\mu k} = \sum_{\ell=1}^{N+2} C_{\ell k}\ f_\ell(\mu), \quad \frac{1}{E_\mu(t)} = \sum_{k=1}^{m} e^{-t/\tau_k'} \sum_{\ell=1}^{N+2} C_{\ell k}\ f_\ell(\mu) \tag{10}$$

in which for $\mu > 1$:

$$f_1(\mu) = 1,\ f_2(\mu) = \mu,\ f_3(\mu) = \mu^2, \ldots, f_{N+2}(\mu) = 0 \tag{11}$$

and for $\mu = 1$ (with $J_{\mu k} = C_{N+2\ k}$):

$$f_1(\mu) = \cdots = f_{N+1}(\mu) = 0,\ f_{N+2}(\mu) = 1. \tag{12}$$

Substituting for $J_{\mu k}$ in Eq.(7) finally provides

$$J(t,t') = \sum_{k=1}^{m} \sum_{\ell=1}^{N+2} C_{\ell k} H_{\ell k}(t,t'), \quad H_{\ell k}(t,t') = \sum_{\mu=1}^{n} f_\ell(\mu)\phi_{\mu k}(t,t'). \tag{13}$$

The condition of optimum fit of given data may now be expressed as

$$\Phi = \sum_\alpha \sum_r \{\sum_k \sum_\ell C_{\ell k} H_{\ell k}(t,t') - \tilde{J}_{r\alpha}\}^2 + \sum_k \sum_\ell w_{\ell k} C_{\ell k}^2 + \Phi_1 = \min. \tag{14}$$

where $\tilde{J}_{r\alpha}$ = given data points $J(t_\alpha' + \bar{t}_r, t_\alpha')$, for discrete ages $t' = t_\alpha'$ (α = 1,2,3,...) and discrete elapsed times $t - t' = t_r$ (r = 1,2,3,...). The data points must be distributed sufficiently densely in log-time over the whole range of $(t - t')$ values and t' values considered.

The weights, $w_{\ell k}$, which are to be suitably chosen, are introduced in Eq. (14) in order to suppress the curvatures of the spectrum, which seems to be essentially a random property. A further suitably chosen weight, w_1, must be introduced in Eq.(14) by means of the term

$$\Phi_1 = w_1 \sum_{\ell=1}^{N-1} \sum_{k=2}^{m-1} (C_{\ell\ k-1} - C_{\ell\ k})^2. \tag{15}$$

This term gives a penalty for unsmooth dependence of coefficients $C_{\ell k}$ upon k.

This penalty is necessary because $C_{\ell k}$ determine through Eq.(9) the coefficients $J_{\mu k}$ of the Dirichlet series in Eq.(5), and as is well known (see also (1,2)), the problem of identifying the coefficients of a Dirichlet series from given values of the series has a non-unique (unstable) solution and leads to ill-conditioned systems of equations, unless an additional condition, such as the afore-mentioned smoothness requirements, is imposed. The term $J_{\mu k}$ for $k = m$ and all μ must be left out from the smoothing condition in Eq.(15), for it represents a constant term of the series in Eq.(5) (since $\tau'_m = \infty$), and such a constant term actually stands for a sum of all exponential terms which are beyond the range of the ages of interest in Eq.(5). Coefficients $J_{\mu k}$ for $k = m$ are determined by means of coefficients $C_{\ell m}$, Eq.(9), and consequently these coefficients may not appear in Eq.(15). This is the reason why the upper limit of the sum in Eq.(15) is $m-1$ rather than m.

By imposing the minimizing conditions $\partial\Phi/\partial C_{pq} = 0$, one obtains the following system of $m(N+2)$ linear equations for $C_{\ell k}$:

$$\sum_{\ell=1}^{N+2} \sum_{k=1}^{m} A_{pq\ell k}\, C_{\ell k} = B_{pq} \tag{16}$$

in which
$$A_{pq\ell k} = \sum_{\alpha} \sum_{r} H_{pq_{\alpha r}} H_{\ell k_{\alpha r}} + \delta_{pk}\delta_{q\ell} w_{pq} + W_{pq\ell k}, \tag{17}$$

$$B_{pq} = \sum_{\alpha} \sum_{r} \tilde{J}_{r\alpha} H_{pq_{\alpha r}}, \tag{18}$$

where $H_{pq_{\alpha r}} = H_{pq}(t'_\alpha + \bar{t}_r, t'_\alpha)$, δ_{pq} = Kronecker delta, and

$$\begin{aligned}
&W_{pqpq} = 2w_1, \quad W_{pqp\,q-1} = W_{pqp\,q+1} = -w_1 && \text{for } 2 \le q \le m-2\\
&W_{pqpq} = w_1, \quad W_{pqp\,q+1} = -w_1 && \text{for } q = 1\\
&W_{pqpq} = w_1, \quad W_{pqp\,q-1} = -w_1 && \text{for } q = m-1\\
&W_{pqpq} = 0 && \text{for all other subscripts.}
\end{aligned} \tag{19}$$

In computer programming, the use of standard subroutines for linear equations requires that the quadruple subscripts of matrix $A_{pq\ell k}$ be replaced by double subscripts i,j, i.e., $A_{pq\ell k} = A_{ij}$ where $i = p + (N+1)(q-1)$, $j = \ell + (N+1)(k-1)$. Similarly, the double subscripts of matrix B_{pq} must be replaced by a single subscript i, i.e., $B_{pq} = B_i$. The solution of the equation system, Eq.(16), is obtained from the equation solving subroutine as a singly subscripted matrix C_i, which must then be converted to the double subscripted matrix $C_{\ell k}$. Subsequently, one can express the retardation spectra for various t'-values with the help of Eq.(10).

For the same reason as in the case of relaxation spectra (1), the $E_\mu(t')$-values for the retardation times τ_μ which are greater than about 30t' are physically meaningless, because no significant deformation rate can occur in the corresponding dashpots at age t', regardless of the previous deformation history. The value of these E_μ-moduli at age t' can be arbitrarily changed without any appreciable effect on the response of the chain. In particular, all such $E_\mu(t')$-values can be set equal to ∞. The physically meaningful part of retardation spectrum at age t' therefore ends at $\log \tau_\mu \approx \log (30t')$. The remaining part of the spectrum is meaningless (and is not shown in Fig. 3); but for convenience in programming it need not be eliminated from Eq.(10) for E_μ. Furthermore, those coefficients $J_{\mu k}$ in Eq.(5) for which τ_μ is greater than about $30\tau_k'$ have no effect on the response of the Kelvin chain because they appreciably affect only E_μ-values at and after time $t' = \tau_k'$. Conversely, these coefficients cannot be determined uniquely from test data, and if they were not intentionally related to other $J_{\mu k}$-values by means of Eq.(9), and if zero weights were used in Eq.(14), an ill-conditioned system of equations would result.

Practical Application

Since determination of retardation or relaxation spectra is necessary to enable creep analysis of large structural systems, a FORTRAN IV subroutine is given in the Appendix to facilitate applications. Comments within the subroutine explain the details of its organization as well as the input and output data. (The subroutine computes the spectra for a cost of less than $3, using the machine CDC-6400.)

As a practical example, the given creep data are assumed as the solid lines in Fig. 2. (These lines represent smoothed test data of L'Hermite and Mamillan (5), the same as those used in (1).) The retardation spectra obtained with the help of the subroutine in the Appendix are shown in Fig. 3, using a quintic polynomial for the functions $J_{\mu k}$, Eq. (9). As a check, the creep curves are computed from parameters $C_{\ell k}$ of the spectra, using Eqs.(9),(8) and (7), and the result is indicated by dashed lines in Fig. 2. The deviation from given data is acceptable, i.e. the creep curves are recovered from the retardation spectra with a sufficient accuracy. This proves that the retardation spectra fully characterize creep properties. More accurate results can be obtained with higher order polynomials for $J_{\mu k}$. It is also noteworthy that all E_μ-values are positive at all times, which is necessary because of thermodynamic restrictions. (In the previous formulation this condition has not been satisfied (2).)



Fig.2. Given creep function and creep function integrated from the spectra in Fig. 3.



Fig.3. Retardation spectra obtained from the creep function in Fig. 2.

Acknowledgment

Support by the U.S. National Science Foundation under Grant GK-26030 is gratefully acknowledged.

References

1. Z. P. Bažant, A. Asghari, Computation of Age-Dependent Relaxation Spectra, Cement and Concrete Research, Vol. 4, No. 4 (1974).

2. Z. P. Bažant, S. T. Wu, Dirichlet Series Creep Function for Aging Concrete, J. Engng. Mech. Div., Proc. Am. Soc. of Civil Engrs. 99, 367-387 (1973).

3. Z. P. Bažant, Constitutive Equation for Concrete Creep Based on Thermodynamics of Multiphase Systems, Materials and Structures, 3, 3-36 (1970).

4. Z. P. Bažant, Theory of Creep and Shrinkage in Concrete Structures: A Précis of Recent Developments, Mechanics Today, Vol. 2, Pergamon Press (in press).

5. R. L'Hermite, M. Mamillan and C. Lefèvre, Nouveaux résultats de recherches sur la déformation et rupture du béton, Annales de l'Institut Technique de Bâtiment et des Travaux Publics, Vol. 18, No. 207-208, 325-360 (1965).

APPENDIX-FORTRAN IV SUBROUTINE

```
      SUBROUTINE  KELVIN (NA,JA,NDEC,JDEC,NK,NTERM,MUEX,W,WEI,C,FL,EMU1,
     1 T,TP,NT,TLOG,TPL,TAU,NMU,TAUP)

C COMPUTES TIME-DEPENDENT RETARDATION MODULI OF KELVIN CHAIN FROM GIVEN
C CREEP FUNCTION J. USER MUST SUPPLY SUBROUTINE FUNCTION CREEP WHICH
C COMPUTES J FOR ANY TIME T FROM LOADING AND ANY AGE TP AT LOADING.

C INPUT- TAU(2)=SMALLEST RETARDATION TIME APPROPRIATE FOR GIVEN DATA,
C TP(1)=FIRST AGE AT LOADING, NA=NO. OF AGES TP AT LOADING, JA=NO. OF
C AGES TP PER LOG(10), NDEC=NO.OF DECADES LOG(10) OF TIME T FROM LOADING
C JDEC=NO. OF STEPS PER LOG(10) (JDEC=2 OR 3 IS SUFFICIENT).
C W(I)=CHOSEN WEIGHTS TO SMOOTH THE SPECTRUM, SUITABLE VALUES ARE
C W(I)=(1.,.05,.01,.05,.01,.005,0.) REPEATED NK TIMES. WEI=WEIGHT ON
C SQUARED DIFFERENCES C(JK+1,JL)-C(JK,JL), SMOOTHING THE DEPENDENCE ON
C AGE. SUITABLE VALUE IS WEI=0.1. NK, NTERM, MUEX =NUMBER OF TERMS
C AND EXPONENTS CORRESPONDING TO C AND FL. MUEX(JP)=(0,1,2,3,4,5,0)=
C EXPONENTS TO BE SUPPLIED, NTERM=NO. OF TERMS IN J-SUB-MU-K =2 + DEGREE
C OF POLYNOMIAL.

C OUTPUT- C(JK,JL) AND FL(MU,L)=COEFFICIENTS OF THE EXPRESSION FOR EMU1.
C EMU1(MU,IA)=INVERSE RETARDATION MODULI (COMPLIANCES). T(IT),TP(IA)=
C DISCRETE TIMES FROM LOADING AND AGES AT LOADING, NT, NA=THEIR NUMBERS,
C TLOG,TPL=THEIR LOGARITHMS. TAU(MU), NMU=RETARDATION TIMES AND THEIR
C NUMBER. TAUP(K)=TAU-PRIME-SUB-K.

      DIMENSION  MUEX(7),W(35),C(7,5),FL(9,7),EMU1(9,8),T(21),TP(8),
     1 TLOG(21),TPL(21),TAU(9),TAUP(5),H(35,8,21),A(35,36),ZE(35),
     2 NJ(35),AJ(8),X3(8)

C       CHOOSE DISCRETE TIMES T(IT) FROM LOADING,AGES TP(IA) AT LOADING,
C       RETARDATION TIMES TAU(MU), PARAMETERS TAUP(1).

      DTR=10.**(1./FLOAT(JDEC))
      T(1)=(TAU(2)*.05)*DTR
      NT=NDEC*JDEC

      DO  5  IT=2,NT
    5 T(IT)=T(IT-1)*DTR

      X7=10.**(1./FLOAT(JA))

      DO  9  IA=2,NA
      TP(IA)=TP(IA-1)*X7
    9 TPL(IA)=ALOG10(TP(IA))
      TPL(1)=ALOG10(TP(1))
      TAU(1)=1.E-30
C       THIS MEANS THAT THE FIRST SPRING IS COUPLED WITH NO DASHPOT.

      DO  12  MU=3,NMU
   12 TAU(MU)=TAU(MU-1)*10.
      TAUP(1)=2.*TP(1)

      DO  13  K=2,NK
   13 TAUP(K)=TAUP(K-1)*10.
      TAUP(NK)=1.E30

C       INFINITE TAUP(NK) MEANS THAT THE NK-TH TERM FOR EMU1 IS CONSTANT

      N1=NK*NTERM

C       FIRST COMPUTE AUXILIARY ARRAY H(I,IA,IT)

      DO  50  JQ=1,NK
      JJ=NTERM*(JQ-1)
      DO  50  JP=1,NTERM
      I=JJ+JP

      DO  50  IA=1,NA
      DO  50  IT=1,NT
      SUM=0.

      DO  40  MU=1,NMU
      TMUK=(TAU(MU)*TAUP(JQ))/(TAU(MU)+TAUP(JQ))
      RATIO2=T(IT)/TMUK

      IF(RATIO2-1.E-6)  45,45,46

C       WHEN RATIO2 TENDS TO 0., LIM(RATIO2)=0./0. USE TAYLOR SERIES

   45 PHIP=1.-(.5-.1666666667*RATIO2)*RATIO2
      GO TO 47
   46 PHIP=(1.-EXP(-RATIO2))/RATIO2
   47 PHI=((T(IT)/TAU(MU))*PHIP)*EXP(-TP(IA)/TAUP(JQ))
      IF(MU .EQ.  1)  GO TO  48

      FP=FLOAT(MU**MUEX(JP))

      IF(JP .EQ. NTERM)  FP=0.
      GO TO  39
   48 FP=0.
      IF(JP .EQ. NTERM)  FP=1.
   39 FL(MU,JP)=FP

C       SINCE FP IS ALSO NEEDED FOR COMPUTING EMU,IT IS STORED IN FL

   40 SUM=SUM+FP*PHI

   50 H(I,IA,IT)=SUM

C       NOW COMPUTE COEFF. A,B OF THE SYSTEM OF EQUATIONS
```

```
      DO  60  I=1,N1
      SUM=0.

      DO  55  IA=1,NA
      DO  55  IT=1,NT
   55 SUM=SUM+H(I,IA,IT)*CREEP(T(IT),TP(IA))
      ZE(I)=SUM

      DO  60  J=1,N1
      SUM=0.
      IF(J .EQ. I)  SUM=W(I)

      DO  58  IA=1,NA
      DO  58  IT=1,NT
   58 SUM=SUM+H(I,IA,IT)*H(J,IA,IT)

   60 A(I,J)=SUM

C       ADD WEIGHTS WHICH ENFORCE SMOOTH DEPENDENCE OF C ON TAUP

      NK1=NK-1
      DO  64  JQ=1,NK
      JJ=NTERM*(JQ-1)
      DO  64  JP=1,NTERM
      I=JJ+JP

      IF(JQ .EQ. NK)  GO TO  64
      A(I,I)=A(I,I)+WEI
      IF(JQ .NE. NK1)  GO TO  61
      J=I-NTERM
      A(I,J)=A(I,J)-WEI
      GO TO  64
   61 IF(JQ .NE. 1)  GO TO  62
      J=I+NTERM
      A(I,J)=A(I,J)-WEI
      GO TO  64
   62 A(I,I)=A(I,I)+WEI
      J=I-NTERM
      A(I,J)=A(I,J)-WEI
      J=I+NTERM
      A(I,J)=A(I,J)+WEI
   64 CONTINUE

C       USE A LIBRARY SUBROUTINE TO SOLVE EQUATIONS (NJ=AUXILIARY ARRAY)

      CALL  SO1SP(A,ZE,N1,1,1.E-8,IERR,N1,NJ)

      DO  68  JK=1,NK
      JJ=NTERM*(JK-1)
      DO  67  JL=1,NTERM
      I=JJ+JL
   67 C(JL,JK)=ZE(I)
   68 PRINT  69, JK, (C(JL,JK),  JL=1,NTERM)
   69 FORMAT(6H C(JL,I2,2H)=10E12.4)

      PRINT  76, (TP(IA),  IA=1,NA)
   76 FORMAT(89H0VALUES OF AJ(IT,IA) OF CREEP FUNCTION J COMPUTED (AS A
     1CHECK) FROM COEFFICIENTS C(JL,JK)//5X7HTP(IA)=10E12.4)

      PRINT  78
   78 FORMAT( 39H IT     T(IT)  AJ(IT,1)   AJ(IT,2)  ...)

      DO  90  IT=1,NT
      DO  85  IA=1,NA
      SUM=0.

      DO  80  J=1,N1
   80 SUM=SUM+ZE(J)*H(J,IA,IT)

      X3(IA)=CREEP(T(IT),TP(IA))
   85 AJ(IA)=SUM

      PRINT  94, IT, T(IT), (AJ(IA),  IA=1,NA)
   90 PRINT  95, (X3(IA),  IA=1,NA)

   94 FORMAT(1XI2,F9.3,10E12.4/(12X10E12.4))
   95 FORMAT(6X7HGIVEN  10E12.4/(13X10E12.4))

      PRINT  102
  102 FORMAT(105H0RETARDATION SPECTRA-VALUES OF INVERSE RETARDATION MODU
     1LI EMU1(MU,IA) COMPUTED FROM COEFFICIENTS C(JL,JK))

      PRINT  103, (TAU(MU),  MU=1,NMU)
  103 FORMAT(12H0  TAU(MU)= 9E11.3/5H TIME12X4HMU=17X8HMU=2 ...)


      DO  140  IA=1,NA
      DO  130  MU=1,NMU
      SUM=0.

      DO  120   K=1,NK
      SUM1=0.
      DO  110   L=1,NTERM
  110 SUM1=SUM1+C(L,K)*FL(MU,L)
  120 SUM=SUM+SUM1*EXP(-TP(IA)/TAUP(K))

  130 EMU1(MU,IA)=SUM
      SMM=30.*TP(IA)
      DO  135  MU=1,NMU
      IF(TAU(MU) .GE. SMM)  EMU1(MU,IA)=0.
  135 CONTINUE
  140 PRINT  142, TP(IA), (EMU1(MU,IA),  MU=1,NMU)
  142 FORMAT(1XE10.3,1X(11E11.3))

      RETURN
      END
```